

# ODS STATISTICAL GRAPHICS 101

Wei Cheng, ISIS Pharmaceuticals, Inc., Carlsbad, CA

## ABSTRACT

Beginning with SAS®9, a number of procedures join the Output Delivery System (ODS) family to create statistical graphics automatically. This new facility is referred to as ODS Statistical Graphics. The layout and appearance of ODS Statistical Graphics are controlled by ODS templates and styles. You can use the ODS Graphics Template Language (GTL) to generate statistical graphics of your own design; you can use PROC TEMPLATE to customize the statistical graphics generated by SAS procedures; you can use the “SG” family of procedures to create procedure-quality statistical plots. This paper describes how to use ODS Statistical Graphics to:

- create statistical graphs from SAS procedures;
- customize statistical graphs using ODS templates and ODS styles;
- create statistical graphs using ODS Graphics Template Language;
- create statistical graphs using SG procedures.

## INTRODUCTION

We have enjoyed so many features of the ODS since it was introduced in SAS Version 7, and we rely more and more on the ODS to create summary tables for publication and presentation. However, if we want to generate statistical graphs, we have to save analysis results in output data sets, apply GOPTIONS, or even use the Annotate facility to display them with SAS/GRAPH procedures. If we need to send the graphs to different file formats such as HTML, PDF, or RTF, we have to select a proper device driver. Even if we spend lots of time finding the right statements or graph options, most of the time we’re still not satisfied with the appearance of graphs that we generated. We wish that the SAS system could extend its ODS features to statistical graphs.

In SAS 9, a number of procedures in Base SAS, SAS/STAT, SAS/QC, and SAS/ETS can generate statistical graphs as easily as tables. These graphs are 100% integrated with ODS, including ODS styles and ODS destinations. ODS Statistical Graphics is designed to automate the creation of statistical graphs from SAS procedures, so we can focus more on the analysis. However, Graphics Template Language enables us to modify the graphics template to customize the graphs when modifications are needed. In addition, there are a few new procedures in SAS/GRAPH that use the ODS Statistical Graphics infrastructure to create stand-alone graphs so these graphs can be consistent in their appearance with statistical graphs created by other SAS procedures.

If you have used the ODS to generate tables, you will find that ODS Statistical Graphics are very easy to start with. I’ll show you by example how to use this exciting new functionality.

ODS Statistical Graphics are experimental in SAS 9.1. In SAS 9.2, the ODS Statistical Graphics and Graphics Template Language became production software. All the examples in this paper have been tested in SAS 9.2.

## CREATE STATISTICAL GRAPHS FROM SAS PROCEDURES

In SAS 9.2, about 40 SAS/STAT procedures can produce graphical displays that are template-based. The layout and appearance of ODS Graphics are controlled by ODS templates and styles. They support popular image formats GIF, JPEG, PNG, PS, EPSI, etc. ODS Graphics integrate the graphical displays fully with the tabular output. Graph production is automated by the procedure, and all ODS Graphics are governed by the standard ODS statements such as SELECT and EXCLUDE, just like ODS tables. ODS Graphics can generate different plot types with complex graphical layouts, and they are integrated with common file formats HTML, PDF, RTF, LATEX, etc.

### ODS GRAPHICS BASICS

ODS GRAPHICS statement enables ODS automatic graphic capabilities. The basic syntax is

```
ODS GRAPHICS ON < / RESET
                ANTIALIAS = ON | OFF
                BORDER = ON | OFF
                IMAGEFMT = image-file-type | STATIC
                IMAGEMAP = ON | OFF
                IMAGENAME = filename
                HEIGHT = dimension
                SCALE = ON | OFF
                WIDTH = dimension
                more-options
                >;

                procedures or data steps
```

```
ODS GRAPHICS OFF;
```

Where

ODS GRAPHICS < OFF | ON > turns off / on the automatic ODS graphic generation, by default, ODS Graphics is not enabled;

RESET resets one or more ODS GRAPHICS options to their default settings, it's the same as RESET = ALL. If you want to reset only one option, you must specify RESET = separately for that option, for example, RESET = INDEX resets the index counter that is appended to static image files;

ANTIALIAS controls the use of antialiasing to smooth the components of a graph. With antialiasing, pixels at the edge of an object are set to an intermediate color, which makes smoother and more professional looking graphics;

BORDER specifies whether to draw the graph with a border;

IMAGEFMT specifies the image format to display graphics in ODS output, i.e. GIF, PNG, etc. By default, IMAGEFMT = STATIC and ODS dynamically uses the best quality static image format for the active output destination;

IMAGEMAP controls tooltip generation in the HTML destination. Tooltips is a text box that appears in HTML output when you hover over a part of the plot with your mouse pointer;

IMAGENAME specifies the base image filename; it should not include a file extension, ODS automatically adds the appropriate extension;

HEIGHT specifies the height of the graph. You can specify height in PX(pixels), IN(inches), or CM(centimeters);

SCALE specifies whether the fonts and symbol markers are scaled proportionally with the size of the graph;

WIDTH specifies the width of the graph, you can specify widths in PX(pixels), IN(inches), or CM(centimeters).

### EXAMPLES OF CREATING STATISTICAL GRAPHS FROM SAS PROCEDURES

PROC FREQ, PROC LIFETEST, PROC UNIVARIATE, and PROC REG will be used as examples to illustrate how to generate graphs using ODS Statistical Graphics, the same principle can be adapted to other procedures which support ODS Graphics.

#### EXAMPLE 1: CREATING GRAPHS FROM PROC FREQ

FIGURE 1





FIGURE 2



This example demonstrates how to request a frequency plot (Figure 1) and a plot of cumulative frequencies (Figure 2) for a one-way frequency table. The ODS GRAPHICS statement is specified to request ODS Graphics in addition to the tabular output:

```
ods listing style = statistical;  
  
ods graphics on;  
  
proc freq data = sashelp.heart;  
  tables bp_status;  
run;  
  
ods graphics off;
```

#### EXAMPLE 2: ONLY SELECT A FREQUENCY PLOT

Each statistical procedure that produces ODS Graphics has a PLOTS = option that is used to select graphs and specify some options. The syntax of the PLOTS = option is as follows:

**PLOTS** < (*global-plot-options*)> < = *plot-request* < (*options*) >>

```
ods graphics on;
```

```
proc freq data = sashelp.heart;
  tables bp_status / plots (only) = freqplot;
run;


ods graphics off;
```

The ONLY option is used when you want to see only the plots specifically listed in the plot-request list. After submitting the preceding statements, only Figure 1 will be generated. You can use ODS SELECT to select graphs as well (see Example 15).

Even though PLOTS = options is a common option for all statistical procedures, the specific global plot options, plot requests, and plot options vary across procedures. You can find the specific syntax from the documentation for each procedure that produces ODS Graphics. The following few examples show you some of the plot options.

### EXAMPLE 3: DOT PLOT INSTEAD OF BAR CHART

FIGURE 3



TYPE = plot-option specifies the plot type. TYPE = BARCHART / BAR produces a bar chart, and TYPE = DOTPLOT / DOT produces a dot plot. The default is TYPE = BARCHART.

```
ods graphics on;

proc freq data = sashelp.heart;
```

```

tables bp_status / plots (only) = freqplot (type=dot);
run;

ods graphics off;

```

#### EXAMPLE 4: ORIENTATION OF THE PLOT

ORIENT = plot-option controls the orientation of the plot.

```


ods graphics on;

proc freq data = sashelp.heart;
  tables bp_status / plots (only) = freqplot (type=dot orient = vertical);
run;

ods graphics off;

```

FIGURE 4



ORIENT = VERTICAL places the frequencies or statistic-values on the y-axis and the variable levels on the x-axis. The default orientation is ORIENT = HORIZONTAL for dot plots and ORIENT = VERTICAL for bar charts.

#### EXAMPLE 5: TWO-WAY FREQUENCY BAR CHART

The TABLES statement requests a cross-tabulation table of Sex by Bp\_status and frequency bar chart for this table. The plots are grouped by the row variable, which is the first variable you specify in a two-way table request.


```
ods graphics on;

proc freq data = sashelp.heart;
  tables sex * bp_status / plots (only) = freqplot;
run;

ods graphics off;
```

The default TYPE = BARCHART. You can request a frequency dot plot as well by specifying TYPE = DOT in the plot-option for FREQPLOT. For multi-way tables, PROC FREQ provides a two-way frequency plot for each stratum.

FIGURE 5



EXAMPLE 6: FREQUENCY PLOT LAYOUT - GROUPHORIZONTAL


TWOWAY = plot-options applies to FREQPLOT for two-way and multi-way tables. The default is TWOWAY = GROUPVERTICAL, which produces a grouped plot with a vertical common

baseline. `TWOWAY =GROUPTHORIZONTAL` produces a grouped plot with a horizontal common baseline as in Figure 6.

The `TYPE =` and `ORIENT =` plot-options are available for each `TWOWAY =` layout options.

```
ods graphics on;  
  
proc freq data = sashelp.heart;  
    tables sex * bp_status / plots(only) = freqplot  
                                (twoway = grouphorizontal);  
run;  
  
ods graphics off;
```

**FIGURE 6**



**EXAMPLE 7: FREQUENCY PLOT LAYOUT - STACKED**

`TWOWAY = STACKED` produces stacked frequency plots for two-way tables. In a stacked bar chart, the row frequencies are stacked within each column, and bars corresponding to the column variable values.

```
ods graphics on;
```


```

proc freq data = sashelp.heart;
  tables sex * bp_status / plots (only) = freqplot
                        (twoway = stacked);
run;

ods graphics off;

```

FIGURE 7



EXAMPLE 8: DEVIATION PLOT

DEVIATIONPLOT requests a plot of relative deviations from expected frequencies for a one-way table. The DEVIATIONPLOT is associated with the CHISQ option for a one-way table request.

```

ods graphics on;

proc freq data = sashelp.heart;
  tables bp_status / chisq plots (only) = deviationplot (type = dot);
run;


ods graphics off;

```

Figure 8 shows the deviation plot, which displays the relative deviations from the hypothesized values. The relative deviation for a level is the difference between the observed and hypothesized percentage divided by the hypothesized percentage. The chi-square statistics is significant at the 0.05 level ( $p < 0.0001$ ). This indicates a significant departure from the hypothesized percentages.

The chi-square  $p$ -value can be suppressed by specifying the NOSTATS plot-option. It is displayed in the graph by default.

FIGURE 8



EXAMPLE 9: PRODUCT-LIMIT SURVIVAL FUNCTION ESTIMATE CURVE

The `PLOTS = SURVIVAL <(survival-options)>` in PROC LIFETEST requests that the Product-Limit Survival Function Estimates curve be plotted as shown in Figure 9.

In the following example, WKS is the name of the failure time variable, CENS is the name of the censoring variable, and VAC is the variable that determine the strata levels.

```
ods graphics on;


proc lifetest data = hsv plots = survival;
  time wks * cens (1);
  strata vac;
run;

ods graphics off;
```

By using the PLOTS = option, you can request other statistical graphs such as a panel of plots for each stratum, a plot of the Hall-Wellner bands for all strata, a plot of the equal precision bands of

all strata, a plot of negative log of the estimated survival functions, and a plot of the log of the negative log of the estimated survival functions.

FIGURE 9



**EXAMPLE 10: DISPLAY THE NUMBER OF SUBJECTS AT RISK**

*Survival-option* ATRISK  $\langle = \text{number-list} \rangle$  displays the number of subjects at risk at the given times. The *number-list* identifies the times at which the numbers at risk are displayed. If the *number-list* is not specified, PROC LIFETEST provides a default list.

```
ods graphics on;

proc lifetest data = hsv
    plots = survival(atrisk=0 to 50 by 10);

    time wks * cens (1);
    strata vac;

run;

ods graphics off;
```

FIGURE 10



**EXAMPLE 11: DISPLAY THE P-VALUE**

*Survival-option* TEST displays the p-value of a homogeneity test specified in the STRATA statement. By default, TEST = (LOGRANK WILCOXON) for stratified and trend tests. If more than one test is produced, the test is chosen in the following order: LOGRANK, WILCOXON, TARONE, PETO, MODPETO, FLEMING, and LR.

```
ods graphics on;


proc lifetest data = hsv
    plots = survival(test atrisk=0 to 50 by 10);

    time wks * cens (1);
    strata vac;

run;

ods graphics off;
```

FIGURE 11



**EXAMPLE 12: DISPLAY CONFIDENCE BANDS**

*Survival-option* CB *<=keyword>* displays the confidence bands (i.e. simultaneous confidence intervals) for the survivor functions. You can display both the equal-precision and the Hall-Wellner bands; the default is Hall-Wellner confidence band. There is another *survival-option* CL that requests displaying the pointwise confidence limits for the survivor function.

```
ods graphics on;


proc lifetest data = hsv
    plots = survival(test atrisk=0 to 50 by 10 cb);

    time wks * cens (1);
    strata vac;

run;

ods graphics off;
```

FIGURE 12



**EXAMPLE 13: GRAPH SIZE AND RESOLUTION**

ODS provides options to specify the size and resolution of graphs. You can specify the size of a graph in the ODS GRAPHICS statement by WIDTH= and HEIGHT= options, you can change the resolution with the IMAGE\_DPI= option in any ODS destination statement.

The default dimensions of ODS Graphics are 640 pixels wide and 480 pixels high. The default resolution of graphs created with HTML and LISTING is 100 DPI (dots per inch), whereas the default with RTF is 200 DPI.

```
ods listing image_dpi = 300 style = statistical;

ods graphics on / width = 4 in height = 3 in;

proc lifetest data = hsv
    plots = survival(atrisk=0 to 50 by 10);


    time wks * cens (1);
    strata vac;

run;

ods graphics off;
```

The Product-Limit Survival Estimates curve in Figure 13 is generated by the preceding program; you will notice that the size of the graph has been reduced while the resolution of the graph has been increased.

FIGURE 13



**EXAMPLE 14: HISTOGRAM FROM PROC UNIVARIATE**

Unlike other procedures in this paper, the UNIVARIATE procedure supports ODS Graphics on an experimental basis in SAS 9.2, which means the appearance of graphs and the syntax are subject to change in a future release.

ODS Graphics provides histogram, cdf plot, P-P plot, probability plot, and Q-Q plot in the UNIVARIATE procedure. The following example demonstrates how to request a comparative histogram for the variable WEIGHT.


The NROWS= option specifies the number of rows per panel in the comparative histogram. By default, comparative histograms are displayed in two rows per panel.

```
ods graphics on;

proc univariate data = sashelp.heart;
  class weight_status;
  histogram weight / nrows = 3 ;
run;

ods graphics off;
```

FIGURE 14



EXAMPLE 15: GRAPH WITH TOOL TIPS IN HTML

This example illustrates how to request graphics in HTML that will display tool tips. When you specify the HTML destination and the `IMAGEMAP = ON` option in the `ODS GRAPHICS` statement, an image map of coordinates for tool tips is generated along with the HTML output file.

```
ods html body = "univariate.html";

ods graphics on / imagemap = on;

ods select histogram;

ods noproctitle;

title "Graphs with Tool Tips in HTML";


proc univariate data = sashelp.heart;
  class weight_status;
  histogram weight / nrows = 3 ;
run;

ods graphics off;

ods html close;
```

The ODS SELECT statement chooses only the histogram be included in the HTML file. Moving the mouse over a histogram displays a tool tip with detailed information for the corresponding bar, as shown in Figure 15. Graphics with tool tips are supported for the HTML destination only.

FIGURE 15



#### EXAMPLE 16: REGRESSION ANALYSIS PLOTS

The REG procedure in the following program produces a summary panel, as shown in Figure 16.1, which consists of eight fit diagnostics plots; a plot of the residuals versus the regressor in the model, as shown in Figure 16.2; a scatter plot of the input data overlaid with the fitted regression line, confidence band, and prediction limits, as shown in Figure 16.3.

```
ods graphics on;

proc reg data = sashelp.heart plots (maxpoints = none);
  model weight = height;
run; quit;

ods graphics off;
```

To display the plots individually instead of the default summary panel, specify the PLOTS(UNPACK) option in the PROC REG statement.

MAXPOINTS = *number* | NONE option specifies that plots with elements that require processing more than number points are suppressed. The default is MAXPOINTS = 5000. This cutoff is ignored if you specify MAXPOINTS = NONE.

MAXPOINTS = NONE is specified in the above program because there are more than 5000 number points that require processing. By default, plots won't be generated.

FIGURE 16.1




FIGURE 16.2





FIGURE 16.3




## ODS STYLES FOR STATISTICAL GRAPHICS

ODS styles control the appearance of graph elements such as fonts, colors, markers, fitted lines, confidence bands, etc. You can specify a style using the `STYLE=` option in an ODS destination statement as in the following example:


```
ods destination file = "filename.ext" style = statistical;
```

You can apply any style provided by the ODS to your graphs. A few of them are recommended for use with ODS Graphics: `STATISTICAL`, `ANALYSIS`, `DEFAULT`, `LISTING`, `JOURNAL`, `JOURNAL2`, and `RTF`. The appearance of these styles is demonstrated in the following graphs (except `STATISTICAL`). Style `STATISTICAL` is used for all the examples in this paper; its dominant colors are blue, creamy gray, and white, with sans-serif fonts.


**STYLE = ANALYSIS: SIMILAR TO STATISTICAL, WHOSE DOIMANT COLOR IS TAN**




STYLE = DEFAULT: DOMINANT COLORS ARE BLUE, GRAY, AND WHITE, WITH BOLD SANS-SERIF FONTS




STYLE = JOURNAL: A BLACK AND WHITE STYLE WITH FILLED AREAS AND SANS-SERIF FONTS




STYLE = JOURNAL2: SIMILAR TO JOURNAL, BUT WITH EMPTY AREAS



STYLE = RTF: DOMINANT COLORS ARE BLUE, WHITE, AND BLACK, WITH TIMES ROMAN FONTS



STYLE = LISTING: SIMILAR TO DEFAULT, BUT WITH A WHITE BACKGROUND



## CUSTOMIZE STATISTICAL GRAPHS USING ODS TEMPLATES AND ODS STYLES

The layout and details of graphs produced with ODS Graphics are controlled by graph template definitions (or graph definition for short). The graph definitions are written in the graph template language. The default graph templates supplied by SAS should suffice in most common applications of ODS Graphics. However, ODS Graphics enables you to modify the default template to customize the graphics when modifications are needed. The overall appearance of ODS Graphics is coordinated by style definitions, like ODS tables. Style definitions specify features such as fonts, sizes, color scheme, and line styles. You can create your own style definition or modify a style definition supplied by SAS to customize the appearance of ODS Graphics.

The default graph template Stat.Lifetest.Graphics.ProductLimit will be used to demonstrate how to customize the layout and appearance of statistical graphics with templates.

### EXAMPLE 17: CUSTOMIZING LAYOUT OF PRODUCT LIMIT SURVIVAL ESTIMATE CURVE

Graph definition Stat.Lifetest.Graphics.ProductLimit generates a graph of the product-limit survival function estimate versus survival time, as shown in Figure 11. If we want to change the default label "Survival Probability" of the y-axis to "My Special Label of Survival Probability", and move the legend box which specifies p-value and censored symbol from top right to bottom left, and add grid lines to the y-axis, we can modify the graph template in the following program.

```
ods path work.mytpl(update)
      sashelp.tmplmst(read);
```

```

proc template;
  define statgraph Stat.Lifetest.Graphics.ProductLimitSurvival;
    dynamic NStrata xName plotAtRisk plotCensored plotCL plotHW plotEP
      labelCL labelHW labelEP maxTime StratumID classAtRisk plotBand
      plotTest GroupName yMin Transparency SecondTitle TestName pValue;
    BeginGraph;
      [SNIP]
      layout overlay /
      xaxisopts=(shortlabel=XNAME offsetmin=.05
      linearopts=(viewmax=MAXTIME))
      yaxisopts=(griddisplay = on
        label="My Special Label of Survival Probability"
        shortlabel="Survival" linearopts=(viewmin=0 viewmax=1
        tickvaluelist=(0 .1 .2 .3 .4 .5 .6 .7 .8 .9 1.0)));
      [SNIP]
      if (PLOTCESTORED)
        if (PLOTTEST)
          layout gridded / rows=2 autoalign=(bottomleft) border=true
          BackgroundColor=GraphWalls:Color Opaque=true;
          entry "+ Censored";
          if (PVALUE < .0001)
            entry TESTNAME " p " eval (PUT(PVALUE, PVALUE6.4));
          else
            entry TESTNAME " p=" eval (PUT(PVALUE, PVALUE6.4));
          endif;
          endlayout;
        else
          layout gridded / rows=1 autoalign=(bottomleft) border=true
          BackgroundColor=GraphWalls:Color Opaque=true;
          entry "+ Censored";
          endlayout;
        endif;
      else
        if (PLOTTEST)
          layout gridded / rows=1 autoalign=(bottomleft) border=true
          BackgroundColor=GraphWalls:Color Opaque=true;
          if (PVALUE < .0001)
            entry TESTNAME " p " eval (PUT(PVALUE, PVALUE6.4));
          else
            entry TESTNAME " p=" eval (PUT(PVALUE, PVALUE6.4));
          endif;
          endlayout;
        endif;
      endif;
    endlayout;
  endif;
EndGraph;
end;
run;

ods listing style = statistical;

ods graphics on;

proc lifetest data = hsv
  plots = survival(test atrisk=0 to 50 by 10);

  time wks * cens (1);

```

```

strata vac;


run;

ods graphics off;

```

The result of the change, after submitting the modified template definition and the preceding program, is displayed in Figure 17:

FIGURE 17



**EXAMPLE 18: CUSTOMIZING APPEARANCE OF PRODUCT LIMIT SURVIVAL ESTIMATE CURVE**

You can modify a style to customize the appearance of ODS Graphics in the same way as modifying a style to customize the appearance of ODS tables. The following example shows how to enlarge the fonts for the axis value (tick mark) labels, axis labels, and titles.

```

proc template;
  define style mystatistical;
  parent = styles.statistical;
  replace GraphFonts /
    'GraphDataFont' = ("<sans-serif>, <MTsans-serif>",9pt)
    'GraphUnicodeFont' = ("<MTsans-serif-unicode>",9pt)
    'GraphValueFont' = ("<sans-serif>, <MTsans-serif>",14pt)

```

```

'GraphLabelFont' = ("<sans-serif>, <MTsans-serif>",14pt)
'GraphFootnoteFont' = ("<sans-serif>, <MTsans-serif>",10pt,italic)
'GraphTitleFont' = ("<sans-serif>, <MTsans-serif>",16pt,bold)
'GraphAnnoFont' = ("<sans-serif>, <MTsans-serif>",10pt);
end;
run;

ods listing style = mystatistical;

ods graphics on;

proc lifetest data = hsv
    plots = survival(test atrisk=0 to 50 by 10);

    time wks * cens (1);
    strata vac;


run;

ods graphics off;

```

Submit the preceding program to display the Product-Limit survival curve created with your modified style, as shown in Figure 18:

FIGURE 18



## CREATE STATISTICAL GRAPHS USING ODS GRAPHICS TEMPLATE LANGUAGE

Starting in SAS 9, a new template type STATGRAPH, which describes the layout and appearance of a graph, was added to the ODS template type family. A stand alone graphics template can be defined, unassociated with any procedure. It's useful when a procedure does not produce a graph that you would like to see. You can define the graphics template, compile the template, use SAS procedures to create output data sets with analyzed results, and finally produce the graph by feeding the data to the graphics template via the new SAS 9.2 SAS/GRAPH procedure SGRENDER.

### ODS GTL BASICS

In SAS 9, ODS GTL is a new language for defining graphs. The syntax is part of PROC TEMPLATE:

```
PROC TEMPLATE;
    DEFINE STATGRAPH name-of-graph-definition;
        BEGINGRAPH </ option(s)>;
            <statgraph-global-statements >
            statgraph-layout-block
            <statgraph-global-statements >
        ENDGRAPH;
    END;
RUN;
```

When the above graph definition is submitted, an output template named name-of-graph-definition is compiled and stored in the template store. To produce the graph, the compiled template must be bound to a data set through PROC SGRENDER and directed to an ODS destination as in the following example. The visual appearance of the graph is controlled by the ODS style in effect:

```
ods destination file = "filename.ext" style = statistical;

proc sgrender data = dataname
    template = "name-of-graph-definition";
run;
```

The following two examples demonstrate how to use ODS GTL to create statistical graphics templates and generate statistical graphics from PROC SGRENDER.

#### EXAMPLE 19: SCATTER PLOT OVERLAID WITH THE FITTED REGRESSION LINE

```
proc template;
    define statgraph simplereg;
        begingraph;
            entrytitle "Simple Scatter Plot with Regression Line";
            layout overlay;
                scatterplot x = height y = weight;
                regressionplot x = height y = weight;
            endlayout;
        endgraph;
    end;
run;

ods listing style = statistical;
```


```
ods graphics on;

proc sgrender data = sashelp.class template = simplereg;
run;

ods graphics off;
```

ENTRYTITLE statement displays a title. LAYOUT OVERLAY statement builds a composite from one or more *statgraph-statements*. Submit the preceding program to display the scatter plot with regression line, as shown in Figure 19:

FIGURE 19



EXAMPLE 20: LOESS FIT BY GENDER

```
proc template;
  define statgraph myplot;
    begingraph;
      entrytitle "Loess Fit by Gender";
      layout overlay;
        scatterplot x = height y = weight / group = sex name = 'sex';
        loessplot x = height y = weight / goup = sex name = 'loess';
        layout gridded / autoalign = (topleft);
        discretelegend 'sex' 'loess' / opaque = false border = true
          merge = true;
    endgraph;
  end;
run;
```

```

        endlayout;
    endlayout;
    endgraph;
end;
run;

ods listing style = statistical;

ods graphics on;

proc sort data = sashelp.class out = class;
    by sex;
run;


proc sgrender data = class template = myplot;
run;

ods graphics off;

```

LOESSPLOT statement Creates a fitted loess curve computed from input data. LAYOUT GRIDDED statement assembles the results of nested *statgraph-statements* into a grid. After submitting the preceding program, a scatter plot with fitted Loess curves grouped by gender is displayed in Figure 20:

FIGURE 20



## CREATE STATISTICAL GRAPHS USING SG PROCEDURES

In SAS 9.2, SAS/GRAPH introduces a family of new procedures to create stand-alone graphs that use the ODS Statistical Graphics infrastructure and are designed to use ODS styles. The new "SG" family of procedures includes SGPLOT, SGPANEL, and SGSCATTER. Graphs created by these procedures are consistent in their appearance, as well as the way they are produced and managed.

### EXAMPLE 21: SGPLOT

SGPLOT creates plots (scatter, fit line, box, dot, series, step, needle, and band) and charts (histogram, bar, line, normal curve, and kernel density estimate) with overlay capabilities.


```
ods graphics on;

proc sgplot data = sashelp.class;
  scatter x = height y = weight / group = sex name = "scatter";
  loess x = height y = weight / group = sex clm
      clmtransparency = 0.6 nomarkers;

  keylegend "scatter";
run;

ods graphics off;
```

FIGURE 21



### EXAMPLE 22: SGPANEL


SGPANEL creates paneled plots and charts driven by classification variables. The classification variables that drive the paneling are specified in the PANELBY statement. There are two types of paneling supported: panel and lattice. When the lattice type is used as in the following example, you are limited to two classification variables for the two dimensions.

```
ods graphics on;

proc sgpanel data = sashelp.heart;
  panelby weight_status bp_status / layout = lattice;
  reg x = height y = weight / cli;
run;

ods graphics off;
```

FIGURE 22



### EXAMPLE 23: SGSCATTER – PLOT


SGSCATTER creates comparative scatter plots and scatter plot matrices, with the capability to overlay confidence ellipses and fit lines. This procedure has only three statements: PLOT, COMPARE, and MATRIX. The PLOT statement creates independent plots for each Y \* X pair in a panel. It is not limited to scatter plot. You can request regression, loess, or penalized B-spline fits of the scatter plots.

```
ods graphics on;

proc sgscatter data = sashelp.heart;
  plot (diastolic systolic) * (weight height) / group = weight_status;
run;

ods graphics off;
```

FIGURE 23



### EXAMPLE 24: SGSCATTER – COMPARE


COMPARE statement in PROC SCATTER creates a comparison matrix that crosses a list of X variables with a list of Y variables. The plot pairs share axes. The MATRIX statement in PROC SCATTER creates a scatter plot matrix of a list of variables.

```
ods graphics on;

proc sgscatter data = sashelp.heart;
  compare x = (diastolic systolic)
         y = weight;
run;

ods graphics off;
```

FIGURE 24



## CONCLUSION

As the old truism states, “One picture is worth a thousand words.” Graphics are an invaluable tool in modern statistical analysis. ODS Statistical Graphics gives us convenient access to commonly used graphics for a specific analysis. With simple ODS statements and options, we can generate all types of graphical displays and complex graphical layouts with pre-designed style and appearance automatically from SAS procedures. We can direct the graphics to destinations such as RTF, HTML, and PDF. We can easily generate images in popular formats such as GIF, PNG, and JPEG. We can choose from several ODS styles, or we can customize the graphics. ODS Statistical Graphics integrates high-quality graphics along with tabular output for many statistical procedures.

## REFERENCES

1. SAS Institute, Inc. "Statistical Graphics Using ODS" *SAS/STAT User's Guide, Chapter 21*, Cary, NC: SAS Institute, Inc.
2. SAS Institute, Inc. "The FREQ Procedure" *SAS/STAT User's Guide, Chapter 35*, Cary, NC: SAS Institute, Inc.
3. SAS Institute, Inc. "The LIFETEST Procedure" *SAS/STAT User's Guide, Chapter 49*, Cary, NC: SAS Institute, Inc.
4. SAS Institute, Inc. "The UNIVARIATE Procedure" *Base SAS Procedures Guide, Chapter 4*, Cary, NC: SAS Institute, Inc.
5. SAS Institute, Inc. "The REG Procedure" *SAS/STAT User's Guide, Chapter 73*, Cary, NC: SAS Institute, Inc.
6. SAS Institute, Inc. "SAS/GRAPH: Graph Template Language Reference" <<http://support.sas.com/rnd/datavisualization/gtl/>> (Feb 22, 2008).
7. Rodriguez, Robert N. (2004), "An Introduction to ODS for Statistical Graphics in SAS® 9.1" *Proceedings of the 29th Annual SAS Users Group International (SUGI) Conference*.
8. Rodriguez, Robert N. (2006), "Creating Statistical Graphics in SAS 9.2: What Every Statistical User Should Know" *Proceedings of the 31th Annual SAS Users Group International (SUGI) Conference*.
9. Cartier, Jeff (2006), "A Programmer's Introduction to the Graphics Template Language" *Proceedings of the 31th Annual SAS Users Group International (SUGI) Conference*.
10. Cartier, Jeff (2005), "The Power of the Graphics Template Language" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
11. Heath, Dan (2007), "SAS/GRAPH Procedures for Creating Statistical Graphics in Data Analysis" *Proceedings of the SAS Global Forum 2007*.
12. Cartier, Jeff (2007), "Using ODS Styles with SAS/GRAPH" *Proceedings of the SAS Global Forum 2007*.
13. Cheng, Wei (2005), "SAS 8 Is Seen in the Rearview Mirror: Generate Statistical Graphics Using ODS in SAS® 9" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
14. Cheng, Wei (2007), "ODS Statistical Graphics for Clinical Research" *Proceedings of the SAS Global Forum 2007*.

## ACKNOWLEDGMENTS

I would like to thank Bob Rodriguez, Sanjay Matange, Jeff Cartier, and Dan Heath of SAS Institute for their documentation and help.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:  
Wei Cheng,  
Isis Pharmaceuticals, Inc.,  
1896 Rutherford Rd, Carlsbad, CA 92008  
Email: [wcheng@isisph.com](mailto:wcheng@isisph.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.